


Tutorial for the program


Galva

by

F5BU, Jean-Paul Gendner
(2019-12 V3.00)




Galva = command interpreter



It is essentially a drawing software. The plots are made, not directly with the help of the mouse, but with the help of commands which allow a very high precision of the plots.

The coordinates



To position the plots on the screen and the printer, the commands use the coordinates of the points.


Cartesian coordinates X,Y, or polar coordinates if useful.

By default, the lower left corner of the screen (or sheet) has the coordinates X=0, Y=0, and, for example, a point of coordinates 50,30 (the center of the black cross in the figure) is a point which is 50 mm to the right of the left edge and 30 mm above the lower edge of the window / sheet.

The coordinates of the mouse are displayed at the bottom right of the graphical part, as well as the cursor line in the code.

Note: Mouse coordinates can be entered in the code by a simple Ctrl-double-click.

Commands examples



The graph on the left is drawn according to the commands entered on the right:

- The "**Trait**" [Trait = Line in English] command draws the line from 0,0 to 20,30 mm, 2 mm thick and blue.
- The "**EpaisT**" [Epaisseur = Thickness] command sets a default line thickness of 1 mm for the following commands.
- The "**Couleur**" [Couleur = Color] command sets the green color as the default color for the following commands.
- The "**Cercle**" [Cercle = Circle] command draws, centered on the end of the blue line (= last plotted point, because the coordinates are not specified), a circle of 15 mm radius.
- The "**Texte**" [Texte = Text] command writes Text starting from 20,5 mm, in red, twice the default size and in bold.

Galva

First steps

Pressing F1 gives access to help. It is recommended to start by reading the introduction, or at least to fly over it.

Menu File, New, gives something like:

' F1 => Help

' F2 => current command input mask (novelty Version 2.5)

'ZoomIni = 1.5

CT = 50,50,n

'RefCur = %CTx,%CTy (novelty Version 2.6)

The **CT** (Temporary Center) command defines the point that will have the 0, 0 coordinates thereafter.

➔ Add following the command: Trait = 0,0,80,60

who becomes **Trait** = 0,0,80,60 when pressing Enter (CR) or moving the cursor to another line.

➔ Press F4 or Visualize Menu to see the result of your first graphic command.

Galva

You have drawn a line.

You find that it is not the right color:

- Insert **Couleur** = blue **before** the line **Trait** ...

You find the line too thin:

- Insert **Epaist** = 1mm **before** the line **Trait** ...

You want to draw another line.


Enter the beginning of command **Trait** = , or only **Trait** then click on F2 (with the cursor on the line of the **Trait** command).

An Input Mask appears, with a description of the command and each parameter, which greatly facilitates their input.

Input Mask by F2

Command: Trait = 0,0,80,60

Trait = : Draws one or several lines as a broken line



[Counterclockwise rotations angle around 0,0]

x0 Coordinate 0

y0 Coordinate 0

x1 Coordinate 80

y1 Coordinate 60

[Lines: thickness_in_mm mm or thickness_in_pixels]

[Type of ends (R=Ronded by default or C=Squared)]

[Length 1 -> First line length (0)]

[Length 2 -> space length]

[Length 3 -> second line length]

[Color (Alt-C) or N]

[x2 Coordinate]

[y2 Coordinate]

[x3 Coordinate]

[y3 Coordinate]

[...]

As for all commands, the xi, yi can be written:
x[R], y[R] An R indicating a relative value with respect to the current point.
Angle_Dir*, Length in polar coordinates (Angle direction and Length).
For the Direction angles (see the drawing).

Valider - OK Abandon - Cancel

Input Mask by F2 (continuation)

The title of the mask describes the command.
Then, there is a description of each parameter.

Parameters between brackets [...] are optional and can be omitted, ie left empty. The other parameters are in bold and must be filled in (except for rare exceptions).

A value in parenthesis (...) indicates the value used by default.
There is none for the **Trait** command.

Depending on the command, additional informations are given in the lower part of the input mask.

For the **Trait** command, the different ways of entering coordinates are summarized. This informations are very important because they are valid for all commands.

Coordinate types

The x_i and y_i are the coordinates of the different points.

Each x and y coordinate can be entered in one of three following forms:

- **absolute** in mm ex.: 50, 100 ; -20, 0,
- **relative** in mm by adding an R or r after the value
ex.: -40r,30r ; 10 R, 0r,
- **polar** (ex.: 110°,50) with the angle in degrees and the length in mm (obligatorily for an x , y couple).

In addition, each value can be entered directly in numerical form or by means of a mathematical expression with or without variable(s) (a variable name starts with %):

ex.: $50 \cdot \sin(30) - 1$ or $\%A \cdot \%x + \%B$, or $\text{Log}((2 \cdot \%y - \%dy) / \%Y\text{Max})$, ...

The best is to try.

Galva

Suppose now that we want to change the color of one of the lines.

For this, we can of course place before a **Couleur** command, but this one modifies the color by default and applies to all the following commands.

It is also possible to use the Color parameter of the **Trait** command, which we saw in the input mask window. Just click F2 again (or enter:

Trait = ,,50,100,,green)

Since the color parameter comes after a Lines Thickness parameter, which defines ... the thickness of the line or lines, it is necessary to leave a parameter empty before putting the color, in case you do not use the input mask.

- **To have all the explanations concerning a command, it is enough to put the cursor on the name of the command and to push Shift F1, and, if necessary, find the right command.**
- **For a list of all commands press F2 with the cursor on a blank line.**
- **To access a summary of all commands and functions, with links to details, push Shift-F2.**

Galva

Enter now, at the end of the code, the command:

Cercle = ,,5 and a circle appears (if you do not forget to do F4!) centered on the end of the last line drawn, since the x and y coordinates of the center have been omitted. F2 allows to edit the command.

Each command leaves the pointer at a defined position, and you just have to omit the x and y values of the next command to use that position.

A **Cercle** = 50,100,5,,,green command, draws a circle centered on position 50,100 (possibly relative to the last **CT** command) in green. By using the input mask (click on F2) to modify the parameters of the command, it is not useful to know that the color is the sixth parameter.

Galva

Finally, enter Texte, click F2 and enter the parameters to get:


Texte = -25r,,*2bc,How beautiful

To make the text "How beautiful" appear, with the center of the text 25 mm left and at the same height as the center of the green circle (last drawn).

- «*2» defines characters twice as large as default;
- «b» specifies a writing in bold;
- «c» specifies that the coordinates are those of the center of the text.

Galva

Here a code near of the previous tests with besides comments and a "?"= command, which makes it possible to record the coordinates of the current point in variables (here %X and %Y), usable later: here in **Trait** commands and a **Texte** command that displays the coordinates of the point.



```
CT = 50,50,n      ' Temporary Center in 50,50
                  ' n -> no mark
RefCur = %CTx,%CTy ' Reference coordinates for the
                  ' display of the mouse coordinates
Couleur = blue    ' default color
EpaisT = 1.2mm    ' default thickness of the lines
Trait = 0,0,80,60 ' draw a line from 0,0 to 80,60
? = %X,%Y         ' memorization of coordinates
Trait = ,,50,100,2mm,red ' absolute coordinates
Trait = %X,%Y,50,100,,vert ' absolutes with expression
Trait = 80,60,-40r,30r,,noir ' draw the black line
Trait = %X,%Y,110°,50 ' draw a blue line
Cercle = ,,5      ' draw the blue circle
Cercle = 50,100,5,,,vert ' draw the green circle
' The following command writes "How beautiful" in blue
Texte = -25r,,, *2bc,How beautiful
' The following command writes "80, 60" next to this point
Texte = %X+5,%Y,magenta,*2by,%X, %Y
```

Galva

Let's also see how to make a simple graduated scale for a galvanometer.

Let's do again Menu File, New, to start a new program. Then remove the ", n" from the **CT** = 50,50,n command, and press F4 to view the change. A small "+" sign appears at the 0,0 temporary coordinates.

Now enter the special **Arc** command which defines the arc for a galvanometer scale. A click on F2 opens the Input Mask with the description of the parameters. Let's complete the command to get:
Arc = 90-30,90+30,60

An arc of circle of radius 60mm was drawn ... of course, but it is necessary to press F4 to see it :-).

Galva

90 is the angle for the middle of the scale and the deviation of 30 ° left and right. Of course, we could have written 60,120,60, but since operations are allowed, this makes it to better display the relevant values, which are then easier to modify later.

This also allows us to go further in this direction and to show the interest of the variables.

Enter the command **%A = 30** before the **Arc** command and replace the **Arc** command by **Arc = 90-%A,90+%A,60**. It is now sufficient to modify the value of %A by another, for example 35, to obtain an arc of 55° to 125°.

Command: Arc = 90-%A,90+%A,60

Arc = : Define the deviation for an arced scale

Beginning Angle of the arc [-360 à 360] 90-%A

End Angle of the arc [-360 à 360] 90+%A

[Counterclockwise rotations angle around 0,0 (0)]

Radius of the arc (>0) 60

[Color (Alt-C) or N to not draw the arc]

[? to get the equivalent Cercle command in comment if Decal>0]

A Arc, Droite or DroiteV command is obligatory before using a Grad, Grad1, Val*, Val1* or Sect command.
For the angles (see the drawing):
For a curved scale with an offset center, the Decal command must precede the Arc command..
Rotation Angle turns the arc around the point 0,0, and acts on the commands Grad *, Val * and Sect.


Angles Winkel

90 45 135 180 225 270 315 0

Valider - OK Abandon - Cancel

Galva

Now you have to add graduations and values. Below, an example of commented code.



```

CT = 50,50          ' Temporary Center
RefCur = %CTx,%CTy ' Reference coordinates for display

Epaist = 0.15 mm    ' Default Thickness of the lines
TailleP = *2g        ' 2 times the default size of the characters
Police = arial       ' Default Font

%A = 35              ' Variable for the half deviation in °
Decal = -20,rouge    ' Vertical offset of the arc axis
                    ' relative to the axis of the galvanometer

Arc = 90-%A, 90+%A, 60 ' Definition of the galvanometer arc
'Droite = 0,75        ' Definition for an horizontal scale
'DroiteV = 0,75       ' Definition for a vertical scale
'Inversion            ' to invert the position of the graduations
                    ' and Values

Exposant = 1.2        ' Modifies the linearity of the graduations
Grad = 50,1           ' Graduations for 50 intervals1,
                    ' Length = 1 mm

Grad = 10,2,,,.3mm    ' 10 grad. of 2 mm and 0.3mm thickness
Grad = 5,3,,,.5mm,bleu ' 5 graduations of 3mm, 0.5 and blue
ValR = ,5,-10,40$V,6,,magenta ' Values from -10 to 40
                    ' in magenta

' click on F2 for the Input Mask of the command on the cursor line
  
```

¹ : For the commands **Grad** and **Val[{R;D}]**, it is necessary to indicate the number of wished intervals, and they draw $n + 1$ graduations or values.

Galva

Since version 2.5, external commands facilitate a number of tasks. For example, the last 5 lines of the preceding code and the commands **TailleP** and **Police** can be replaced by the call to the command **CGE_E** as shown by the input mask below.

Command: GCE_E = ,,5\$10\$50,3\$2\$1,5mm\$3mm,bleu\$noir\$noir,5,-10,40,6,V,magenta,Arial,2g,1.2

GCE_E = : Draw a scale with values and up to 3 levels of graduations

[x Coordinate (0)]	
[y Coordinate (0)]	
Number of main graduations (5)	5
[Number of level 2 graduations (10)]	10
[Number of level 3 graduations (0)]	50
Length of main graduations (2)	3
[Length of level 2 graduations (1)]	2
[Length of level 3 graduations (.5)]	1
Thickness of the main graduations (.3mm)	.5mm
[Thickness of level 2 graduations (.2mm)]	.3mm
[Thickness of level 3 graduations (.1mm)]	
Color of main graduations (blue)	bleu
[Color of level 2 graduations (black)]	noir
[Color of level 3 graduations (black)]	noir
Number of values to display (5)	5
Start value (0)	-10
Full Scale Value (10)	40
Distance between values and scale (4)	6
[Text for Unit after last value ()]	V
[Color of text (black)]	magenta
[Font ()]	Arial
[Font Size (Style) (1.3 g)]	2g
[Exponent defining the scale linearity (1)]	1.2
[Start of area for scale processing (0)]	
[End of area for scale processing (1)]	
[Nothing, D or R without space ()]	
[R] Values rotation angle ()	
[Line type, rounded by default, or C for squares ()]	

Allows to draw scales easily.
With the last field empty, the Val command is used and values are written horizontally, with R it is ValR and values are written perpendicular to the radius, and with D it is ValD and values are written perpendicular to the offsetted radius.
The angle of rotation is relative, not absolute, if an R is placed in front.

Valider - OK Abandon - Cancel

There are 5 main intervals (thus 5 + 1 graduations), 10 "graduations" of levels 2 and 50 of level 3. Then, for each level of graduation it is necessary to specify the length of the marks, their thickness and their color.

In the same way we ask the display in magenta of 5 (+1) values ranging from -10 to 40 and placing them at 6mm from the scale. A "V" is added after the last value. Etc.

Galva

- pressing the F1 key opens the help.
- with the cursor on a command line, pressing F2 opens the Input Mask describing the command and all parameters.
- The words of the commands are in blue once the line is validated by changing the line or viewing the result by F4.
- An apostrophe " ' " at the beginning of the line transforms the line into a comment that takes the green color.
- Likewise, with the exception of the **Text** and **TextC** commands, everything after an apostrophe in a line is comment.
- A "**Stop**" or "!" command stops the execution of the code.
- A "**ListeVar**" command displays the contents of all the variables at the location of the code where it is.
- a line ending by " _" (space-underscore) continues in the next line.
- the decimal separator in the **Galva** code is the dot (".").
- the dimensions are in mm and the angles in degrees.

Galva

Now you may do some tests and consult the help.

Above all, do not hesitate to use one of the many example files provided, quickly presented in the help and viewable in the GalvCmd*.pdf and Galva_Cmd*.pdf files.

You can also consult the tutorial Tuto_Galva...(in french or in english) for the realization of dials of galvanometers or Cadran_TSF... (in french) for dials of old radios.

Good fun with Galva

Jean-Paul, F5BU